Pensieve header: The Logos in the ucw order (user-responsibility degrees).

Implementing g_1

```
ϵ /: ϵ^2 = 0;
PBWRule = {u → 1, c → 2, w → 3};
B[U@c, U@w] = -(B[U@w, U@c] = U@w);
B[U@u, U@c] = -(B[U@c, U@u] = U@u);
B[U@w, U@u] = -(B[U@u, U@w] = b U[] - 2 ϵ U@c);
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[]; UU[l_, r___] := U[l] ** UU[r];
U_i_[ℰ_] := ℰ /. {b → b_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
```

```
B[x_, x_] = 0;
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := If[x ≺ y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := If[x ≤ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := U@xx ** (U@xn ** U@yy);
```

```
U[l___, x_^n_, r___] := U[l, Sequence @@ Table[x, {n}], r];
U[l___, 1, r___] := U[l, r];
```

```
bc_i_ := b_i U[] - ϵ U[c_i];
a_i_,j_ := bc_i ** U@c_j + U@u_i ** U@w_j;
```

```
UExp[ℰ_, n_] := Module[{t = U[], k}, U[] + Sum[(t = t ** ℰ)/k!, {k, n}]] // Simp
```

```
ToDegree[n_][ℰ_] := Simp[ℰ] /. a_. x_U ⧴ Normal[Series[a, {ħ, 0, n}]] * x
```

```
LBasis[n_Integer] := LBasis[Range[n]];
LBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
        Table[{U@c_i, U@u_i, U@w_i, ϵ U@c_i, ϵ U@u_i, ϵ U@w_i}, {i, S}],
        Table[{U[u_i, w_j], ϵ U[u_i, w_j],
          ϵ U @@ Sort@{c_i, c_j}, ϵ U[u_j, c_i], ϵ U[c_i, w_j]}, {i, S}, {j, S}],
        Table[{ϵ U[u_j, c_i, w_k], ϵ U @@ Sort@{u_i, u_j, w_k}, ϵ U @@ Sort@{u_i, w_j, w_k}},
         {i, S}, {j, S}, {k, S}],
        Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
BLBasis[n_Integer] := BLBasis[Range[n]];
BLBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
        Table[{U@c_i, ϵ U@c_i}, {i, S}],
        Table[{U[u_i, w_j], ϵ U[u_i, w_j], ϵ U @@ Sort@{c_i, c_j}}, {i, S}, {j, S}],
        Table[{ϵ U[u_j, c_i, w_k]}, {i, S}, {j, S}, {k, S}],
        Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
𝕆[n_, poly_, specs___] := Module[{vs, us},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :→ (l /. x_i_ :→ x_s));
  Total[
   CoefficientRules[Normal@Series[poly, {ħ, 0, n}], vs] /. (p_ → c_) :→ c UU @@ (us^p)
  ]
 ]
```

Testing gl

```
Table[x → B[U@x, a_{1,2}], {x, {c_1, c_2, u_1, u_2, w_1, w_2, c_3, u_3, w_3}}] // Column
```

$c_1 \to U[u_1, w_2]$
$c_2 \to -U[u_1, w_2]$
$u_1 \to \epsilon U[u_1, c_2]$
$u_2 \to b_2 U[u_1] - b_1 U[u_2] - 2 \epsilon U[u_1, c_2] + \epsilon U[u_2, c_1]$
$w_1 \to -b_1 U[w_2] + 2 \epsilon U[c_1, w_2] - \epsilon U[c_2, w_1]$
$w_2 \to b_1 U[w_2] - \epsilon U[c_1, w_2]$
$c_3 \to 0$
$u_3 \to 0$
$w_3 \to 0$

```
B[a_{1,3}, a_{2,3}] + B[a_{1,2}, a_{2,3}] + B[a_{1,2}, a_{1,3}]
```

0

```
{t1 = B[a_{1,2}, a_{1,3}], t2 = ∈ U@c_3 ** a_{1,2} - ∈ U@c_2 ** a_{1,3}, t1 - t2} // Expand // Column
```

$-\in U[u_1, c_2, w_3] + \in U[u_1, c_3, w_2]$
$-\in U[u_1, c_2, w_3] + \in U[u_1, c_3, w_2]$
0

```
{t1 = B[a_{1,3}, a_{2,3}], t2 = bc_2 ** a_{1,3} - bc_1 ** a_{2,3}, t1 - t2} // Expand // Column
```

$b_2 U[u_1, w_3] - b_1 U[u_2, w_3] - \in U[u_1, c_2, w_3] + \in U[u_2, c_1, w_3]$
$b_2 U[u_1, w_3] - b_1 U[u_2, w_3] - \in U[u_1, c_2, w_3] + \in U[u_2, c_1, w_3]$
0

```
{t1 = a_{1,3} ** a_{2,4} - a_{1,4} ** a_{2,3},
  t2 = a_{1,3} ** bc_2 ** U@c_4 - a_{2,3} ** bc_1 ** U@c_4 - a_{1,4} ** bc_2 ** U@c_3 + a_{2,4} ** bc_1 ** U@c_3,
  t1 - t2
 } // Column
```

$-b_2 U[u_1, c_3, w_4] + b_2 U[u_1, c_4, w_3] + b_1 U[u_2, c_3, w_4] - b_1 U[u_2, c_4, w_3] +$
$\in U[u_1, c_2, c_3, w_4] - \in U[u_1, c_2, c_4, w_3] - \in U[u_2, c_1, c_3, w_4] + \in U[u_2, c_1, c_4, w_3]$
$-b_2 U[u_1, c_3, w_4] + b_2 U[u_1, c_4, w_3] + b_1 U[u_2, c_3, w_4] - b_1 U[u_2, c_4, w_3] +$
$\in U[u_1, c_2, c_3, w_4] - \in U[u_1, c_2, c_4, w_3] - \in U[u_2, c_1, c_3, w_4] + \in U[u_2, c_1, c_4, w_3]$
0

```
LBasis[2]
```

$\{U[], \in U[], U[c_1], U[c_2], U[u_1], U[u_2], U[w_1], U[w_2], \in U[c_1], \in U[c_2], \in U[u_1], \in U[u_2],$
$\in U[w_1], \in U[w_2], U[u_1, w_1], U[u_1, w_2], U[u_2, w_1], U[u_2, w_2], \in U[c_1, c_1], \in U[c_1, c_2],$
$\in U[c_1, w_1], \in U[c_1, w_2], \in U[c_2, c_2], \in U[c_2, w_1], \in U[c_2, w_2], \in U[u_1, c_1], \in U[u_1, c_2],$
$\in U[u_1, w_1], \in U[u_1, w_2], \in U[u_2, c_1], \in U[u_2, c_2], \in U[u_2, w_1], \in U[u_2, w_2], \in U[u_1, c_1, w_1],$
$\in U[u_1, c_1, w_2], \in U[u_1, c_2, w_1], \in U[u_1, c_2, w_2], \in U[u_1, u_1, w_1], \in U[u_1, u_1, w_2],$
$\in U[u_1, u_2, w_1], \in U[u_1, u_2, w_2], \in U[u_1, w_1, w_1], \in U[u_1, w_1, w_2], \in U[u_1, w_2, w_2],$
$\in U[u_2, c_1, w_1], \in U[u_2, c_1, w_2], \in U[u_2, c_2, w_1], \in U[u_2, c_2, w_2], \in U[u_2, u_2, w_1],$
$\in U[u_2, u_2, w_2], \in U[u_2, w_1, w_1], \in U[u_2, w_1, w_2], \in U[u_2, w_2, w_2], \in U[u_1, u_1, w_1, w_1],$
$\in U[u_1, u_1, w_1, w_2], \in U[u_1, u_1, w_2, w_2], \in U[u_1, u_2, w_1, w_1], \in U[u_1, u_2, w_1, w_2],$
$\in U[u_1, u_2, w_2, w_2], \in U[u_2, u_2, w_1, w_1], \in U[u_2, u_2, w_1, w_2], \in U[u_2, u_2, w_2, w_2]\}$

```
bas = LBasis[2];
Table[B[x, y] + B[y, x], {x, bas}, {y, bas}] // Flatten // Union
```

{0}

```
bas = LBasis[2]; Timing[
 Table[
    {x, y, z} = xyz;
    Simp[B[B[x, y], z] + B[B[y, z], x] + B[B[z, x], y]],
    {xyz, Subsets[bas, {3}]}
   ] // Flatten // Union
]
```

{28.5, {0}}

Infinitesimal Spinner Relations

```
s_{i_} := b_i/2 U[] - ∈ U[c_i]
```

$a_{1,2}$

$b_1 U[c_2] - \epsilon U[c_1, c_2] + U[u_1, w_2]$

$B[a_{1,2}, s_1]$

$\epsilon U[u_1, w_2]$

$B[a_{1,2}, s_2]$

$-\epsilon U[u_1, w_2]$

$B[a_{1,2}, s_1 + s_2]$ // Simplify

0

$B[a_{1,3}, a_{2,3}]$

$b_2 U[u_1, w_3] - b_1 U[u_2, w_3] - \epsilon U[u_1, c_2, w_3] + \epsilon U[u_2, c_1, w_3]$

$\text{Simplify}[b_1 UU[u_1, w_3] - b_1 UU[u_1, w_3] + \epsilon UU[c_1, u_1, w_3] - \epsilon UU[u_1, c_1, w_3]]$

$\epsilon U[u_1, w_3]$

Testing bi-local exponential relations in g_0

1. The Yang-Baxter Element.
2. c Relations.

$B[U[c], U[u, w]]$

0

$\mathbb{O}\left[3, e^{\hbar (\gamma c_0 + \beta u_0)}, \{u_0, c_0\} \to 1\right]$

$$U[] + \gamma \hbar U[c_1] + \beta \hbar U[u_1] + \frac{1}{2} \gamma^2 \hbar^2 U[c_1, c_1] + \beta \gamma \hbar^2 U[u_1, c_1] + \frac{1}{2} \beta^2 \hbar^2 U[u_1, u_1] + \frac{1}{6} \gamma^3 \hbar^3 U[c_1, c_1, c_1] + \frac{1}{2} \beta \gamma^2 \hbar^3 U[u_1, c_1, c_1] + \frac{1}{2} \beta^2 \gamma \hbar^3 U[u_1, u_1, c_1] + \frac{1}{6} \beta^3 \hbar^3 U[u_1, u_1, u_1]$$

$\mathbb{O}\left[3, e^{\hbar (\gamma c_0 + e^{-\hbar \gamma} \beta u_0)}, \{c_0, u_0\} \to 1\right]$

$$U[] + \gamma \hbar U[c_1] + \left(\beta \hbar - \beta \gamma \hbar^2 + \frac{1}{2} \beta \gamma^2 \hbar^3\right) U[u_1] + \frac{1}{2} \gamma^2 \hbar^2 U[c_1, c_1] + \left(\beta \gamma \hbar^2 - \beta \gamma^2 \hbar^3\right) (U[u_1] + U[u_1, c_1]) + \left(\frac{\beta^2 \hbar^2}{2} - \beta^2 \gamma \hbar^3\right) U[u_1, u_1] + \frac{1}{6} \gamma^3 \hbar^3 U[c_1, c_1, c_1] + \frac{1}{2} \beta \gamma^2 \hbar^3 (U[u_1] + 2 U[u_1, c_1] + U[u_1, c_1, c_1]) + \frac{1}{2} \beta^2 \gamma \hbar^3 (2 U[u_1, u_1] + U[u_1, u_1, c_1]) + \frac{1}{6} \beta^3 \hbar^3 U[u_1, u_1, u_1]$$

$\text{With}\left[\{n = 7\}, \text{Simp}\left[\mathbb{O}\left[n, e^{\hbar (\gamma c_0 + \beta u_0)}, \{u_0, c_0\} \to 0\right] == \mathbb{O}\left[n, e^{\hbar (\gamma c_0 + e^{-\hbar \gamma} \beta u_0)}, \{c_0, u_0\} \to 0\right]\right]\right]$

True

$\text{With}\left[\{n = 7\}, \text{Simp}\left[\mathbb{O}\left[n, e^{\hbar (\gamma c_0 + \hbar \beta w_0)}, \{w_0, c_0\} \to 0\right] == \mathbb{O}\left[n, e^{\hbar (\gamma c_0 + \hbar e^{\hbar \gamma} \beta w_0)}, \{c_0, w_0\} \to 0\right]\right]\right]$

True

3. u, w, e^u, e^w.

```
With[{n = 10},
 Simp[B[U@w, UExp[γ U@u, n]] + (b γ U[] - 2 γ ∈ U[c] + ∈ γ^2 U[u]) ** UExp[γ U@u, n]]]
```

$$\left(\frac{b\,\gamma^{11}}{3\,628\,800}-\frac{\gamma^{11}\in}{362\,880}\right)U[u,u,u,u,u,u,u,u,u,u]-\frac{\gamma^{11}\in U[u,u,u,u,u,u,u,u,u,u,c]}{1\,814\,400}+\frac{\gamma^{12}\in U[u,u,u,u,u,u,u,u,u,u,u]}{3\,628\,800}$$

```
With[{n = 7},
 Simp[B[U@u, UExp[γ U@w, n]] - (b γ U[] - 2 γ ∈ U[c] - ∈ γ^2 U[w]) ** UExp[γ U@w, n]]]
```

$$\left(-\frac{b\,\gamma^{8}}{5040}+\frac{\gamma^{8}\in}{720}\right)U[w,w,w,w,w,w,w]+\frac{\gamma^{8}\in U[c,w,w,w,w,w,w,w]}{2520}+\frac{\gamma^{9}\in U[w,w,w,w,w,w,w,w]}{5040}$$

4. M_{uw} Relations.

```
Muw[γ_, n_] := Expand[Sum_{k=0}^{n} γ^k/k! UU[u^k, w^k]]
```

```
Muw[γ, 5]
```

$$U[]+\gamma U[u,w]+\frac{1}{2}\gamma^2 U[u,u,w,w]+\frac{1}{6}\gamma^3 U[u,u,u,w,w,w]+\frac{1}{24}\gamma^4 U[u,u,u,u,w,w,w,w]+\frac{1}{120}\gamma^5 U[u,u,u,u,u,w,w,w,w,w]$$

```
𝕆[5, e^(ħ δ u0 w0), {u0, w0} → 1]
```

$$U[]+\delta\hbar U[u_1,w_1]+\frac{1}{2}\delta^2\hbar^2 U[u_1,u_1,w_1,w_1]+\frac{1}{6}\delta^3\hbar^3 U[u_1,u_1,u_1,w_1,w_1,w_1]+\frac{1}{24}\delta^4\hbar^4 U[u_1,u_1,u_1,u_1,w_1,w_1,w_1,w_1]+\frac{1}{120}\delta^5\hbar^5 U[u_1,u_1,u_1,u_1,u_1,w_1,w_1,w_1,w_1,w_1]$$

```
With[{n = 12}, Muw[γ, n] ** U[u] -
    Expand@((1 - b γ - 2 γ ϵ) U[u] + 2 γ ϵ UU[c, u] - (1 - b γ)^-1 γ^2 ϵ U[u, u, w]) **
     Muw[γ, n]] // Simplify // Expand
```

$$-\left(\left(b\,\gamma^{13}\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(479\,001\,600\,(-1+b\,\gamma)\right)\right)+$$
$$\left(b^2\,\gamma^{14}\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(479\,001\,600\,(-1+b\,\gamma)\right)+$$
$$\left(\gamma^{13}\,\epsilon\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(39\,916\,800\,(-1+b\,\gamma)\right)-$$
$$\left(b\,\gamma^{14}\,\epsilon\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(39\,916\,800\,(-1+b\,\gamma)\right)+$$
$$\left(\gamma^{13}\,\epsilon\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,c,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(239\,500\,800\,(-1+b\,\gamma)\right)-$$
$$\left(b\,\gamma^{14}\,\epsilon\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,c,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(239\,500\,800\,(-1+b\,\gamma)\right)-$$
$$\left(\gamma^{14}\,\epsilon\,U[u,u,u,u,u,u,u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w,w,w,w,w,w]\right)\Big/\left(479\,001\,600\,(-1+b\,\gamma)\right)$$

```
With[{n = 7}, Simplify[B[U[w], Muw[γ, n]] + b γ Muw[γ, n] ** U[w]]]
```

$$2\,\gamma\,\epsilon\,U[c,w]+\gamma^2\,\epsilon\,U[u,w,w]+2\,\gamma^2\,\epsilon\,U[u,c,w,w]+$$
$$\gamma^3\,\epsilon\,U[u,u,w,w,w]+\gamma^3\,\epsilon\,U[u,u,c,w,w,w]+\frac{1}{2}\gamma^4\,\epsilon\,U[u,u,u,w,w,w,w]+$$
$$\frac{1}{3}\gamma^4\,\epsilon\,U[u,u,u,c,w,w,w,w]+\frac{1}{6}\gamma^5\,\epsilon\,U[u,u,u,u,w,w,w,w,w]+$$
$$\frac{1}{12}\gamma^5\,\epsilon\,U[u,u,u,u,c,w,w,w,w,w]+\frac{1}{24}\gamma^6\,\epsilon\,U[u,u,u,u,u,w,w,w,w,w,w]+$$
$$\frac{1}{60}\gamma^6\,\epsilon\,U[u,u,u,u,u,c,w,w,w,w,w,w]+\frac{1}{120}\gamma^7\,\epsilon\,U[u,u,u,u,u,u,w,w,w,w,w,w,w]+$$
$$\frac{1}{360}\gamma^7\,\epsilon\,U[u,u,u,u,u,u,c,w,w,w,w,w,w,w]+$$
$$\frac{1}{5040}b\,\gamma^8\,U[u,u,u,u,u,u,u,w,w,w,w,w,w,w,w]$$

```
With[{n = 7}, Simplify[
   (∂_α Muw[γ[α], n]) - γ'[α]/(1 - b γ[α]) Muw[γ[α], n] ** U[u, w]
  ]]
```

$$\frac{1}{5040\,(-1+b\,\gamma[\alpha])}$$
$$\gamma[\alpha]\,\Big(10\,080\,\epsilon\,U[u,c,w]+\gamma[\alpha]\,\Big(5040\,\epsilon\,U[u,u,w,w]+10\,080\,\epsilon\,U[u,u,c,w,w]+$$
$$\gamma[\alpha]\,\Big(5040\,\epsilon\,U[u,u,u,w,w,w]+5040\,\epsilon\,U[u,u,u,c,w,w,w]+$$
$$\gamma[\alpha]\,\Big(2520\,\epsilon\,U[u,u,u,u,w,w,w,w]+1680\,\epsilon\,U[u,u,u,u,c,w,w,w,w]+$$
$$\gamma[\alpha]\,\Big(840\,\epsilon\,U[u,u,u,u,u,w,w,w,w,w]+420\,\epsilon\,U[u,u,u,u,u,c,w,w,w,w,w]+\gamma[\alpha]\,\Big(210\,\epsilon\,U[u,u,u,u,u,u,w,w,w,w,w,w]+$$
$$84\,\epsilon\,U[u,u,u,u,u,u,c,w,w,w,w,w,w]+\Big(42\,\epsilon\,U[u,u,u,u,u,u,u,w,w,w,w,w,w,w]+14\,\epsilon\,U[u,u,u,u,u,u,u,c,w,w,w,w,w,w,w]+$$
$$U[u,u,u,u,u,u,u,u,w,w,w,w,w,w,w,w]\Big)\,\gamma[\alpha]\Big)\Big)\Big)\Big)\Big)\Big)\,\gamma'[\alpha]$$

```
Λ = With[
    {vs = {1, ħ u₀, w₀, ħ² u₀², w₀², ħ² u₀² w₀, ħ u₀ w₀², ħ² u₀² w₀², c₀, ħ c₀ u₀, c₀ w₀, ħ c₀ u₀ w₀, ħ u₀ w₀}},
    Table[a_i, {i, Length@vs}].vs
   ] /. {
    a₁ → 0 a₁ - μ^-1 ħ b₁ δ² - 2 μ^-2 α β δ ħ² b₁ - 1/2 μ^-3 α² β² ħ³ b₁,
    a₂ → 0 a₂ + 2 μ^-2 β δ + μ^-3 α β² ħ,
    a₃ → 0 a₃ + 2 μ^-2 α δ ħ + μ^-3 α² β ħ²,
    a₄ → 0 a₄ + β² δ μ^-3 (1 + ħ b₁ δ / 2),
    a₅ → 0 a₅ + ħ² α² δ μ^-3 (1 + ħ b₁ δ / 2),
    a₆ → 0 a₆ + 3 β δ² μ^-3 (1 + 2/3 ħ b₁ δ),
    a₇ → 0 a₇ + 3 α δ² ħ μ^-3 (1 + 2/3 ħ b₁ δ),
    a₈ → 0 a₈ + 2 δ³ μ^-3 (1 + 3/4 ħ b₁ δ),
    a₉ → 0 a₉ + 2 δ + 2 μ^-1 α β ħ,
    a₁₀ → 0 a₁₀ + 2 β δ μ^-1,
    a₁₁ → 0 a₁₁ + 2 α δ ħ μ^-1,
    a₁₂ → 0 a₁₂ + 2 δ² μ^-1,
    a₁₃ → 0 a₁₃ + 4 δ² μ^-2 (1 + 1/2 ħ b₁ δ) + 4 ħ α β δ μ^-3 (1 + 1/2 ħ b₁ δ)
   };
With[{n = 8}, Simp[
  𝕆[n, ⅇ^(ħ (α w₀+β u₀+δ u₀ w₀)), {w₀, u₀} → 1] -
   𝕆[n, μ^-1 (1 + ħ ϵ μ^-1 Λ) ⅇ^(ħ (-ħ b₁ α β+α w₀+β u₀+δ u₀ w₀)/μ) /. μ → 1 + ħ b₁ δ, {u₀, c₀, w₀} → 1]
 ]]
```

0

```
Simplify[Λ /. {ħ → 1, b₁ → b, u₀ → u, w₀ → w, c₀ → c}]
```

$$\frac{1}{2\mu^3}\Big(-b\,\alpha^2\beta^2+u^2\beta^2\delta\,(2+b\,\delta)+w^2\delta\,(\alpha+u\,\delta)\,\big(\alpha\,(2+b\,\delta)+u\,\delta\,(4+3\,b\,\delta)\big)-4\,b\,\alpha\,\beta\,\delta\,\mu+4\,c\,\alpha\,\beta\,\mu^2-2\,b\,\delta^2\mu^2+4\,c\,\delta\,\mu^3+2\,u\,\beta\,\big(\alpha\,\beta+2\,\delta\,\mu\,(1+c\,\mu)\big)+2\,w\,\Big(\alpha^2\beta+2\,\alpha\,\delta\,\big(u\,\beta\,(2+b\,\delta)+\mu\,(1+c\,\mu)\big)+u\,\delta^2\big(u\,\beta\,(3+2\,b\,\delta)+2\,\mu\,(2+b\,\delta+c\,\mu)\big)\Big)\Big)$$

```
(Collect[Simplify[μ³ Λ /. {ħ → 1, b₁ → b, u₀ → u, w₀ → w, c₀ → c}],
    {c, u, w}, C[Simplify[#]] &] // Expand) /. C[x_] :> x
```

$$\frac{1}{2}w^2\alpha^2\delta\,(2+b\,\delta)+\frac{1}{2}u^2\beta^2\delta\,(2+b\,\delta)+u\,w^2\alpha\,\delta^2\,(3+2\,b\,\delta)+u^2w\,\beta\,\delta^2\,(3+2\,b\,\delta)+\frac{1}{2}u^2w^2\delta^3\,(4+3\,b\,\delta)+2\,c\,w\,\alpha\,\delta\,\mu^2+2\,c\,u\,\beta\,\delta\,\mu^2+2\,c\,u\,w\,\delta^2\mu^2+2\,u\,w\,\delta\,(2+b\,\delta)\,(\alpha\,\beta+\delta\,\mu)+2\,c\,\mu^2\,(\alpha\,\beta+\delta\,\mu)+w\,\alpha\,(\alpha\,\beta+2\,\delta\,\mu)+u\,\beta\,(\alpha\,\beta+2\,\delta\,\mu)-\frac{1}{2}b\,\big(\alpha^2\beta^2+4\,\alpha\,\beta\,\delta\,\mu+2\,\delta^2\mu^2\big)$$

oldΛ = $2\,c\,w\,\alpha\,\delta\,\nu + 2\,c\,u\,\beta\,\delta\,\nu + 2\,c\,u\,w\,\delta^2\,\nu + u\,w^2\,\alpha\,\delta^2\,\nu^3 - \frac{1}{2}\,b\,u^2\,w^2\,\delta^4\,\nu^3 + \frac{1}{2}\,w^2\,\alpha^2\,\delta\,(2 + b\,\delta)\,\nu^3 - u^2\,w\,\beta\,\delta^2\,(1 + 2\,b\,\delta)\,\nu^3 - \frac{1}{2}\,u^2\,\beta^2\,\delta\,(2 + 3\,b\,\delta)\,\nu^3 + 2\,c\,(\delta + \alpha\,\beta\,\nu) - 2\,b\,u\,w\,\delta^2\,\nu^2\,(\delta + \alpha\,\beta\,\nu) + w\,\alpha\,\nu^2\,(2\,\delta + \alpha\,\beta\,\nu) - u\,\beta\,(1 + 2\,b\,\delta)\,\nu^2\,(2\,\delta + \alpha\,\beta\,\nu) - \frac{1}{2}\,b\,\nu\,(2\,\delta^2 + 4\,\alpha\,\beta\,\delta\,\nu + \alpha^2\,\beta^2\,\nu^2)$

$$2\,c\,w\,\alpha\,\delta\,\nu + 2\,c\,u\,\beta\,\delta\,\nu + 2\,c\,u\,w\,\delta^2\,\nu + u\,w^2\,\alpha\,\delta^2\,\nu^3 - \frac{1}{2}\,b\,u^2\,w^2\,\delta^4\,\nu^3 + \frac{1}{2}\,w^2\,\alpha^2\,\delta\,(2 + b\,\delta)\,\nu^3 - u^2\,w\,\beta\,\delta^2\,(1 + 2\,b\,\delta)\,\nu^3 - \frac{1}{2}\,u^2\,\beta^2\,\delta\,(2 + 3\,b\,\delta)\,\nu^3 + 2\,c\,(\delta + \alpha\,\beta\,\nu) - 2\,b\,u\,w\,\delta^2\,\nu^2\,(\delta + \alpha\,\beta\,\nu) + w\,\alpha\,\nu^2\,(2\,\delta + \alpha\,\beta\,\nu) - u\,\beta\,(1 + 2\,b\,\delta)\,\nu^2\,(2\,\delta + \alpha\,\beta\,\nu) - \frac{1}{2}\,b\,\nu\,(2\,\delta^2 + 4\,\alpha\,\beta\,\delta\,\nu + \alpha^2\,\beta^2\,\nu^2)$$

Rescaling the logos as in Projects/OneCo-1606/RescalingTheLogos.nb:

```
nlogos = With[
  {ξ = b/(t - 1), logos = ϵ/(2 μ^3) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) -
       4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 + 4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) +
       2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) + u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ))))},
  logos /. μ → 1 + b δ /. {ϵ -> ξ ϵ, u -> ξ u, β -> ξ^-1 β, δ -> ξ^-1 δ}];
(Collect[Simplify[(1 + (-1 + t) δ)^3 nlogos / ϵ], {c, u, w}, C[Simplify[#]] &] // Expand) /.
  C[x_] :> x
```

$$2\,c\,w\,\alpha\,\delta\,(1 + (-1 + t)\,\delta)^2 + 2\,c\,u\,\beta\,\delta\,(1 + (-1 + t)\,\delta)^2 + 2\,c\,u\,w\,\delta^2\,(1 + (-1 + t)\,\delta)^2 + \frac{1}{2}\,w^2\,\alpha^2\,\delta\,(2 + (-1 + t)\,\delta) + \frac{1}{2}\,u^2\,\beta^2\,\delta\,(2 + (-1 + t)\,\delta) + u\,w^2\,\alpha\,\delta^2\,(3 + 2\,(-1 + t)\,\delta) + u^2\,w\,\beta\,\delta^2\,(3 + 2\,(-1 + t)\,\delta) + \frac{1}{2}\,u^2\,w^2\,\delta^3\,(4 + 3\,(-1 + t)\,\delta) + 2\,c\,(1 + (-1 + t)\,\delta)^2\,(\alpha\,\beta + \delta + (-1 + t)\,\delta^2) + 2\,u\,w\,\delta\,(2 + (-1 + t)\,\delta)\,(\alpha\,\beta + \delta\,(1 + (-1 + t)\,\delta)) + w\,\alpha\,(\alpha\,\beta + 2\,\delta\,(1 + (-1 + t)\,\delta)) + u\,\beta\,(\alpha\,\beta + 2\,\delta\,(1 + (-1 + t)\,\delta)) - \frac{1}{2}\,(-1 + t)\,(\alpha^2\,\beta^2 + 4\,\alpha\,\beta\,\delta\,(1 + (-1 + t)\,\delta) + 2\,\delta^2\,(1 + (-1 + t)\,\delta)^2)$$